Public Function OpenStandardVorlagenWord(plngAdressierungID As Long, Optional strModulzusatz As String, Optional strAblagePfad As String, Optional pstrDatei As String)
On Error GoTo Level_Error

Dim objword As Object
Dim szdoc As String
Dim newdoc As String
Dim varVorlagenpfad As Variant
Dim strAppPathUserData As String
Dim I As Long
Dim pbytparam102 As Byte 'Word Ablagepfad benutzen

Dim pstrVerdankungsbrief As String

    'Pfad für Datenquelle in Variable abfüllen
    strAppPathUserData = GetAppPathUserData
    
    'Versandversion prüfen (neutralisiert oder pro Kontakt)
    If Right(plngAdressierungID, 1) = 1 Then
        'Neutralisierte Adressen (1 x pro Firma)
        Call SendWordTransfer(1, strAppPathUserData, strModulzusatz)
    Else
        '1 Adresse pro Kontakt
        Call SendWordTransfer(0, strAppPathUserData, strModulzusatz)
    End If
    
    'Definiert das Vorlagenverzeichnis
    varVorlagenpfad = GetLizenzPfadDokVorlagen

    Select Case plngAdressierungID
        
        Case 211, 212, 213
            Select Case strModulzusatz
                Case "M06_SpendeEinzel"
                    pstrVerdankungsbrief = Nz(Forms!frm_M6_Buchungsbeleg!cbo_VerdankungsBriefID.Column(1))
                    If pstrVerdankungsbrief = "" Then
                        Call OpenDotAuswahl(varVorlagenpfad & "Spendenverwaltung\Einzelverdankungen\")
                        Exit Function
                    
                    Else
                        szdoc = varVorlagenpfad & "Spendenverwaltung\Einzelverdankungen\" & pstrVerdankungsbrief
                        If existiertDatei(szdoc) = False Then
                            MsgBox "Diese Briefvorlage wurde gelöscht.", vbCritical
                            Exit Function
                        End If
                    End If
               
                Case "M06_SpendeOutputEinzel"
                    Call OpenDotAuswahl(varVorlagenpfad & "Spendenverwaltung\Einzelverdankungen\")
                    Exit Function
                
                Case "M06_SpendeOutputTotal"
                    Call OpenDotAuswahl(varVorlagenpfad & "Spendenverwaltung\Totalisierungen\")
                    Exit Function
                
                Case "M08_KursTn"
                    Call OpenDotAuswahl(varVorlagenpfad & "Anlassverwaltung\Teilnehmer\")
                    Exit Function
                
                Case "M11_Zimmerverwaltung"
                    Call OpenDotAuswahl(varVorlagenpfad & "Zimmerverwaltung\Eigene Vorlagen\")
                    Exit Function
                
                Case Else
                    Call OpenDotAuswahl(varVorlagenpfad & "Eigene Vorlagen\", strAblagePfad)
                    Exit Function
            End Select
            
        Case 221, 222, 223
            szdoc = varVorlagenpfad & "Vorlagen\DBBrief.dotx"
                 
        Case 231, 232, 233
            szdoc = varVorlagenpfad & "Vorlagen\CouvertC5.dotx"
                    
        Case 241, 242, 243
            DoCmd.OpenForm "frm_M0ü_Wordetiketten"
            Exit Function
        
        Case 251, 252, 253
            szdoc = varVorlagenpfad & "Vorlagen\Eti3481S.dotx"
        
        Case 261, 262, 263
            szdoc = varVorlagenpfad & "Vorlagen\Eti3481SP.dotx"
        
        Case 271, 272, 273 'Eigene Formulare
            Select Case strModulzusatz
                Case "M00_ExportSpeicher", "M00_EinzelAdr_PE", "M00_EinzelAdr_FA"
                    Call OpenDotAuswahl(varVorlagenpfad & "Eigene Formulare\")
                    Exit Function
                
                Case "M09_Mitgliederrechnung", "M09_MitgliederMahnung1", "M09_MitgliederMahnung2", "M09_MitgliederMahnung3"
                    Call OpenDotAuswahl(varVorlagenpfad & "Mitglieder\")
                    Exit Function
                
                Case "M11_Zimmerverwaltung"
                    Call OpenDotAuswahl(varVorlagenpfad & "Zimmerverwaltung\Eigene Formulare\")
                    Exit Function
                
                Case Else
                    'Exit Function
            End Select
            
        Case 311, 312, 313
            szdoc = varVorlagenpfad & "Vorlagen\Eti3481B.dotx"
                    
        Case Else:
            MsgBox "Sie haben keine gültige Auswahl getroffen.", vbCritical
            Exit Function
        
    End Select
    
    If IstWordGestartet() = True Then
        Set objword = GetObject(, "Word.Application")    ' Word war schon gestartet
    Else
        Set objword = CreateObject("Word.Application")   ' Word starten
        objword.Documents.Add DocumentType:=0 '0 = con_wdNewBlankDocument
    End If
    
    With objword
        
        'ergänzt 19.02.2011 w/Fehler ab Acc2010 (evt. Acc2007)
        'Wenn Dokumente nacheinander aufgerufen werden, wird ab
        '2. Dokument kein Template mit Word geöffnet
        'deshalb vorgängiges "Documents.add"
        .Documents.Add Template:=szdoc, NewTemplate:=False
                
        'Word - Serienbriefmodus aktivieren und Steuerdatei übergeben...
        .ActiveDocument.MailMerge.OpenDataSource Name:=strAppPathUserData & "Wordtransfer.rtf", LinkToSource:=True, readOnly:=False, AddToRecentFiles:=False
        
        'Einstellung, dass die Feldinhalte und nicht die Seriendruckfelder angezeigt werden
        .ActiveDocument.MailMerge.ViewMailMergeFieldCodes = False
        
[bookmark: _GoBack]        'Zählung der Datensätze wegen automatischem Mischvorgang
        I = DCount("*", "tbl_M1_ExportMerge")
        
        'Mischvorgang prüfen und ausführen
        Select Case plngAdressierungID
            Case Is > 230 Or I = 1
                szdoc = .ActiveDocument
                With .ActiveDocument.MailMerge
                    .Destination = Const_WdSendToNewDocument
                    .Execute
                End With
                
                newdoc = .ActiveDocument
                .Documents("" & szdoc & "").Close Const_WdDoNotSaveChanges
                .Documents("" & newdoc & "").Activate
                
                If I = 1 Then
                    pbytparam102 = Nz(PubFct_Parameter(102), 0) 'Word Ablagepfad benutzen
                    If pbytparam102 = 1 And strAblagePfad <> "" Then
                        If Dir(strAblagePfad, vbDirectory) <> "" Then
                            strAblagePfad = GetLocalPath(strAblagePfad)
                            If Dir(strAblagePfad, vbDirectory) <> "" Then
                                .ChangeFileOpenDirectory strAblagePfad
                                'Vorgabe für Dateiname läuft nicht (28.12.2008) - Ursache nicht gefunden
                                '.activedocument.BuiltInDocumentProperties("Title").Value = "" & pstrDatei & ""
                            End If
                        End If
                    End If
                End If
        End Select
        
        .visible = True
        .Activate
        
        AppActivate (.ActiveDocument)
        
        
    End With
    
Level_Exit:
    Set objword = Nothing
    Exit Function

Level_Error:
    DoCmd.SetWarnings True
    MsgBox Err.Description
    Resume Level_Exit
    
End Function




Public Function SendWordTransfer(pbytKontaktartFlag As Byte, strAppPathUserData As String, ByVal strModulzusatz As String)
On Error GoTo Level_Error

Dim strKunde As String

'Definition pbytKontaktartFlag
'0 = Firmenadressierung auch an SABe, ABT
'1 = Firmenadressierung nur Neutral

    If pbytKontaktartFlag = 0 Then
        Select Case strModulzusatz
            Case "M06_SpendeEinzel", "M06_SpendeOutputEinzel"
                DoCmd.OutputTo acQuery, "qry_M6_SpendenexportWord_lng", "RichTextFormat(*.rtf)", strAppPathUserData & "WordTransfer.rtf", False, ""
           
            Case "M06_SpendeOutputTotal"
                DoCmd.OutputTo acQuery, "qry_M6_SpendenexportWord_krz", "RichTextFormat(*.rtf)", strAppPathUserData & "WordTransfer.rtf", False, ""
            
            Case "M08_KursTn"
                DoCmd.OutputTo acQuery, "qry_M8_Export_KursTn", "RichTextFormat(*.rtf)", strAppPathUserData & "WordTransfer.rtf", False, ""
            
            Case "M09_Mitgliederrechnung"
                DoCmd.OutputTo acQuery, "qry_M9_Fakturierungsrun9Word", "RichTextFormat(*.rtf)", strAppPathUserData & "WordTransfer.rtf", False, ""
            
            Case "M09_MitgliederMahnung1"
                DoCmd.OutputTo acQuery, "qry_M9_Mahnungslauf1Word", "RichTextFormat(*.rtf)", strAppPathUserData & "WordTransfer.rtf", False, ""
            
            Case "M09_MitgliederMahnung2"
                DoCmd.OutputTo acQuery, "qry_M9_Mahnungslauf2Word", "RichTextFormat(*.rtf)", strAppPathUserData & "WordTransfer.rtf", False, ""
            
            Case "M09_MitgliederMahnung3"
                DoCmd.OutputTo acQuery, "qry_M9_Mahnungslauf3Word", "RichTextFormat(*.rtf)", strAppPathUserData & "WordTransfer.rtf", False, ""
            
            Case "M11_Zimmerverwaltung"
                DoCmd.OutputTo acQuery, "qry_M11_ZimmerexportWord", "RichTextFormat(*.rtf)", strAppPathUserData & "WordTransfer.rtf", False, ""
            
            Case "M01_Sesam"
                strKunde = PubFct_LizParameter("Kunde")
                Select Case strKunde
                    Case "VIP"
                        DoCmd.OutputTo acQuery, "qry_M1_Export_WordKinder", "RichTextFormat(*.rtf)", strAppPathUserData & "WordTransfer.rtf", False, ""
                    Case Else
                        DoCmd.OutputTo acQuery, "qry_M1_Export_Word", "RichTextFormat(*.rtf)", strAppPathUserData & "WordTransfer.rtf", False, ""
                End Select
                
            Case Else
                DoCmd.OutputTo acQuery, "qry_M1_Export_Word", "RichTextFormat(*.rtf)", strAppPathUserData & "WordTransfer.rtf", False, ""
                                          
        End Select
    Else
        Select Case strModulzusatz
            Case "M01_Sesam"
                strKunde = PubFct_LizParameter("Kunde")
                Select Case strKunde
                    Case "VIP"
                       DoCmd.OutputTo acQuery, "qry_M1_Export_WordKinder", "RichTextFormat(*.rtf)", strAppPathUserData & "WordTransfer.rtf", False, ""
                    Case Else
                     DoCmd.OutputTo acQuery, "qry_M1_Export_Word_Neutral", "RichTextFormat(*.rtf)", strAppPathUserData & "WordTransfer.rtf", False, ""
                End Select
            
            Case "HTMLMailer"
                    DoCmd.OutputTo acQuery, "qry_M1_HTMLMailer_XLS_Word", "RichTextFormat(*.rtf)", strAppPathUserData & "WordTransfer.rtf", False, ""
                
            Case Else
                DoCmd.OutputTo acQuery, "qry_M1_Export_Word_Neutral", "RichTextFormat(*.rtf)", strAppPathUserData & "WordTransfer.rtf", False, ""
        End Select
    End If
    
Level_Exit:
    Exit Function

Level_Error:
    Select Case Err.Number
        Case 2302
            MsgBox "Sie müssen zuerst das in Bearbeitung befindliche Word-Dokument schliessen.", vbCritical
            Resume Level_Exit
        Case Else
            MsgBox "Fehler " & Err.Number & ": " & Err.Description, vbCritical, con_strAppName
            Resume Level_Exit
    End Select
    
End Function





Public Function IstWordGestartet() As Boolean

'Globale Funktion zur Prüfung, ob Word geladen ist
'
'Rückgaben:
'True: Word ist instanziert
'False: Word ist nicht instanziert

Dim obj As Object

    On Error Resume Next
    Set obj = GetObject(, "Word.Application")
    
    If Err.Number <> 0 Then
        Err.Number = 0
        IstWordGestartet = False
    Else
        IstWordGestartet = True
    End If

    Set obj = Nothing

End Function
